Covariance Structure Approach to Within-Cases

Remember how the data file `grapefruit1.data` looks:

Store	sales1	sales2	sales3
1	62.1	61.3	60.8
2	58.2	57.9	55.1
3	51.6	49.2	46.2
4	53.7	51.5	48.3
5	61.4	58.7	56.6
6	58.5	57.2	54.3
7	46.8	43.2	41.5
8	51.2	49.8	47.9

`proc mixed` wants the data set up as if you planned to ignore within-cases.

store	price	sales
1	1	62.1
1	2	61.3
1	3	60.8
2	1	58.2
2	2	57.9
2	3	55.1
3	1	51.6
3	2	49.2
3	3	46.2
4	1	53.7
4	2	51.5
4	3	48.3
5	1	61.4
5	2	58.7
5	3	56.6
6	1	58.5
6	2	57.2
6	3	54.3
7	1	46.8
7	2	43.2
7	3	41.5
8	1	51.2
8	2	49.8
8	3	47.9

Natural way to read the data is `input store price sales;` Call this a *univariate data read*.

You could still read the data so the 3 sales values are on the same case:

```
input store  price1 sales1
      store2 price2 sales2
      store3 price3 sales3;
```

Call this a *multivariate data read.*

```
/* grapefruit2.sas */
options linesize=79 pagesize=500 noovp formdlim='_';
title "Oneway ANOVA with repeated measures: Covariance Structure Approach";
title2 'Grapefruit data (Kutner et al. 5th ed. Prob 27.6)';

data grape1;
     infile 'grapefruit1.data' firstobs=2; /* Skip the labels */
     input store sales1-sales3;
     label sales1 = 'Sales at Price 1'
           sales2 = 'Sales at Price 2'
           sales3 = 'Sales at Price 3';
     d12 = sales1-sales2; d13 = sales1-sales3; d23=sales2-sales3;

proc glm;
     title3 'Multivariate approach with proc glm';
     model sales1-sales3 = ;
     repeated price / short summary mean;


data grape2;     /* This data set will have 3n cases. */
     set grape1;
     price = 1; sales = sales1; output; /* Output creates a new case. */
     price = 2; sales = sales2; output;
     price = 3; sales = sales3; output;
     keep store price sales;

proc print;
     title3 'Data set with one case per observation';

proc mixed;
     title3 'Proc mixed with unknown covariance structure';
     title4 'Compare F = 29.66 with df = 2, 6';
     class price;
     model sales = price;
     repeated / type=un subject=store r;

proc mixed;
     title3 'Proc mixed with compound symmetry cov. structure and contrasts';
     title4 'Compare F = 49.35 with df = 2, 14';
     class price;
     model sales = price;
     contrast '1vs2' price 1 -1  0;
     contrast '1vs3' price 1  0 -1;
     contrast '2vs3' price 0  1 -1;
     repeated / type=cs subject=store r;
```

The GLM Procedure

Number of Observations Read	8
Number of Observations Used	8

Skipping the univariate output …

The GLM Procedure
Repeated Measures Analysis of Variance

Repeated Measures Level Information

Dependent Variable	sales1	sales2	sales3
Level of price	1	2	3

MANOVA Test Criteria and Exact F Statistics
for the Hypothesis of no price Effect
H = Type III SSCP Matrix for price
E = Error SSCP Matrix

S=1 M=0 N=2

Statistic	Value	F Value	Num DF	Den DF	Pr > F
Wilks' Lambda	0.09185282	29.66	2	6	0.0008
Pillai's Trace	0.90814718	29.66	2	6	0.0008
Hotelling-Lawley Trace	9.88698145	29.66	2	6	0.0008
Roy's Greatest Root	9.88698145	29.66	2	6	0.0008

The GLM Procedure
Repeated Measures Analysis of Variance
Univariate Tests of Hypotheses for Within Subject Effects

Source	DF	Type III SS	Mean Square	F Value	Pr > F
price	2	67.48083333	33.74041667	49.35	<.0001
Error(price)	14	9.57250000	0.68375000		

And skipping the remaining multivariate output.

Recall that the original data file looked like this:

Store	sales1	sales2	sales3
1	62.1	61.3	60.8
2	58.2	57.9	55.1
3	51.6	49.2	46.2
4	53.7	51.5	48.3
5	61.4	58.7	56.6
6	58.5	57.2	54.3
7	46.8	43.2	41.5
8	51.2	49.8	47.9

And here is the output from `proc print`:

Oneway ANOVA with repeated measures: Covariance Structure Approach 9
Grapefruit data (Kutner et al. 5th ed. Prob 27.6)
Data set with one case per observation

Obs	store	price	sales
1	1	1	62.1
2	1	2	61.3
3	1	3	60.8
4	2	1	58.2
5	2	2	57.9
6	2	3	55.1
7	3	1	51.6
8	3	2	49.2
9	3	3	46.2
10	4	1	53.7
11	4	2	51.5
12	4	3	48.3
13	5	1	61.4
14	5	2	58.7
15	5	3	56.6
16	6	1	58.5
17	6	2	57.2
18	6	3	54.3
19	7	1	46.8
20	7	2	43.2
21	7	3	41.5
22	8	1	51.2
23	8	2	49.8
24	8	3	47.9

Grapefruit data (Kutner et al. 5th ed. Prob 27.6)
Proc mixed with unknown covariance structure
Compare F = 29.66 with df = 2, 6

The Mixed Procedure

Model Information

Data Set	WORK.GRAPE2
Dependent Variable	sales
Covariance Structure	Unstructured
Subject Effect	store
Estimation Method	REML
Residual Variance Method	None
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

Class Level Information

Class	Levels	Values
price	3	1 2 3

Dimensions

Covariance Parameters	6
Columns in X	4
Columns in Z	0
Subjects	8
Max Obs Per Subject	3

Number of Observations

Number of Observations Read	24
Number of Observations Used	24
Number of Observations Not Used	0

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	141.05309845	
1	1	86.03317486	0.00000000

Convergence criteria met.

Estimated R Matrix for Subject 1

Row	Col1	Col2	Col3
1	29.6084	33.0114	34.0598
2	33.0114	37.5886	38.7000
3	34.0598	38.7000	40.6255

Covariance Parameter Estimates

Cov Parm	Subject	Estimate
UN(1,1)	store	29.6084
UN(2,1)	store	33.0114
UN(2,2)	store	37.5886
UN(3,1)	store	34.0598
UN(3,2)	store	38.7000
UN(3,3)	store	40.6255

Fit Statistics

-2 Res Log Likelihood	86.0
AIC (smaller is better)	98.0
AICC (smaller is better)	104.0
BIC (smaller is better)	98.5

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
5	55.02	<.0001

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
price	2	7	34.60	0.0002

Compare F = 29.66 with df = 2, 6

Grapefruit data (Kutner et al. 5th ed. Prob 27.6)
Proc mixed with compound symmetry cov. structure and contrasts
Compare F = 49.35 with df = 2, 14

The Mixed Procedure

Model Information

Data Set	WORK.GRAPE2
Dependent Variable	sales
Covariance Structure	Compound Symmetry
Subject Effect	store
Estimation Method	REML
Residual Variance Method	Profile
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

Class Level Information

Class	Levels	Values
price	3	1 2 3

Dimensions

Covariance Parameters	2
Columns in X	4
Columns in Z	0
Subjects	8
Max Obs Per Subject	3

Number of Observations

Number of Observations Read	24
Number of Observations Used	24
Number of Observations Not Used	0

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	141.05309845	
1	1	93.18551859	0.00000000

Convergence criteria met.

Estimated R Matrix for Subject 1

Row	Col1	Col2	Col3
1	35.9408	35.2571	35.2571
2	35.2571	35.9408	35.2571
3	35.2571	35.2571	35.9408

Covariance Parameter Estimates

Cov Parm	Subject	Estimate
CS	store	35.2571
Residual		0.6838

Fit Statistics

-2 Res Log Likelihood	93.2
AIC (smaller is better)	97.2
AICC (smaller is better)	97.9
BIC (smaller is better)	97.3

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	47.87	<.0001

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
price	2	14	49.35	<.0001

Contrasts

Label	Num DF	Den DF	F Value	Pr > F
1vs2	1	14	19.75	0.0006
1vs3	1	14	98.34	<.0001
2vs3	1	14	29.95	<.0001

```
/* monkey2.sas */
options linesize=79 pagesize=100 noovp formdlim='_' nodate;
title 'Primate hippocampal function: Zola-Morgan and Squire, 1990';
title2 'Covariance Structure approach to repeated measures (within-cases)';

data memory2;
     infile 'monkey.data' firstobs=2;
     input monkey $ treatmnt $ week2 week4 week8 week12 week16;
     /* Make 5 "cases" in the data set for each line in the raw data file.
        The  output  command generates a case. */
     id = _n_;
     time = 2;   score = week2;   output;
     time = 4;   score = week4;   output;
     time = 8;   score = week8;   output;
     time = 12;  score = week12;  output;
     time = 16;  score = week16;  output;
     keep monkey treatmnt id time score;

proc print;

proc mixed;
     title3 'Unstructured Covariance Matrix';
     class treatmnt time;
     model score = treatmnt|time;
     repeated / type=un subject=id r;
     /* Could have used subject=monkey, but then monkey must be declared in
        class because it's character-valued. */

/* For comparison, the multivariate approach gave us

                                Num     Den
             Effect              DF      DF    F Value    Pr > F

             treatmnt             1      16       8.08    0.0118
             time                 4      13       0.62    0.6571
             treatmnt*time        4      13       4.12    0.0227

Could justify compound symmetry: Each monkey brings her own special talent
for discrimination learning.  */

proc mixed;
     title3 'Compound Symmetry';
     class treatmnt time;
     model score = treatmnt|time;
     repeated / type=cs subject=id r;

/* Now, since we did not reject the null hypothesis that observations from the
same subject are independent, try an ordinary 2-way ANOVA.  */

proc glm;
     title3 'Ordinary 2-way';
     class treatmnt time;
     model score = treatmnt|time;
```

Obs	monkey	treatmnt	id	time	score
1	Spank	CONTROL	1	2	95
2	Spank	CONTROL	1	4	75
3	Spank	CONTROL	1	8	80
4	Spank	CONTROL	1	12	65
5	Spank	CONTROL	1	16	70
6	Chim	CONTROL	2	2	85
7	Chim	CONTROL	2	4	75
8	Chim	CONTROL	2	8	55
9	Chim	CONTROL	2	12	75
10	Chim	CONTROL	2	16	85
11	Chak	CONTROL	3	2	75
12	Chak	CONTROL	3	4	95
13	Chak	CONTROL	3	8	60
14	Chak	CONTROL	3	12	40
15	Chak	CONTROL	3	16	45

Skipping ...

86	Duncan	TREATED	18	2	65
87	Duncan	TREATED	18	4	55
88	Duncan	TREATED	18	8	55
89	Duncan	TREATED	18	12	80
90	Duncan	TREATED	18	16	75

Unstructured Covariance Matrix

The Mixed Procedure

Model Information

Data Set	WORK.MEMORY2
Dependent Variable	score
Covariance Structure	Unstructured
Subject Effect	id
Estimation Method	REML
Residual Variance Method	None
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

Class Level Information

Class	Levels	Values
treatmnt	2	CONTROL TREATED
time	5	2 4 8 12 16

Dimensions

Covariance Parameters	15
Columns in X	18
Columns in Z	0
Subjects	18
Max Obs Per Subject	5

Number of Observations

Number of Observations Read	90
Number of Observations Used	90
Number of Observations Not Used	0

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	615.76098645	
1	1	606.99648116	0.00000000

Convergence criteria met.

Estimated R Matrix for Subject 1

Row	Col1	Col2	Col3	Col4	Col5
1	109.62	1.1972	-0.2638	5.6006	24.5739
2	1.1972	80.2354	-12.8856	-20.9010	-17.3295
3	-0.2638	-12.8856	102.76	18.9326	-16.3352
4	5.6006	-20.9010	18.9326	97.5649	46.4489
5	24.5739	-17.3295	-16.3352	46.4489	101.14

Covariance Parameter Estimates

Cov Parm	Subject	Estimate
UN(1,1)	id	109.62
UN(2,1)	id	1.1972
UN(2,2)	id	80.2354
UN(3,1)	id	-0.2638
UN(3,2)	id	-12.8856
UN(3,3)	id	102.76

UN(4,1)	id	5.6006
UN(4,2)	id	-20.9010
UN(4,3)	id	18.9326
UN(4,4)	id	97.5649
UN(5,1)	id	24.5739
UN(5,2)	id	-17.3295
UN(5,3)	id	-16.3352
UN(5,4)	id	46.4489
UN(5,5)	id	101.14

Fit Statistics

-2 Res Log Likelihood	607.0
AIC (smaller is better)	637.0
AICC (smaller is better)	644.5
BIC (smaller is better)	650.4

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
14	8.76	0.8458

Covariance Structure approach to repeated measures (within-cases)

Unstructured Covariance Matrix

The Mixed Procedure

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
treatmnt	1	16	8.08	0.0118
time	4	16	0.76	0.5655
treatmnt*time	4	16	5.07	0.0078

Covariance Structure approach to repeated measures (within-cases)
Compound Symmetry

The Mixed Procedure

Model Information

Data Set	WORK.MEMORY2
Dependent Variable	score
Covariance Structure	Compound Symmetry
Subject Effect	id
Estimation Method	REML
Residual Variance Method	Profile
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

Class Level Information

Class	Levels	Values
treatmnt	2	CONTROL TREATED
time	5	2 4 8 12 16

Dimensions

Covariance Parameters	2
Columns in X	18
Columns in Z	0
Subjects	18
Max Obs Per Subject	5

Number of Observations

Number of Observations Read	90
Number of Observations Used	90
Number of Observations Not Used	0

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	615.76098645	
1	1	615.62879226	0.00000000

Convergence criteria met.

Estimated R Matrix for Subject 1

Row	Col1	Col2	Col3	Col4	Col5
1	98.2630	2.9038	2.9038	2.9038	2.9038
2	2.9038	98.2630	2.9038	2.9038	2.9038
3	2.9038	2.9038	98.2630	2.9038	2.9038
4	2.9038	2.9038	2.9038	98.2630	2.9038
5	2.9038	2.9038	2.9038	2.9038	98.2630

Covariance Parameter Estimates

Cov Parm	Subject	Estimate
CS	id	2.9038
Residual		95.3592

Fit Statistics

-2 Res Log Likelihood	615.6
AIC (smaller is better)	619.6
AICC (smaller is better)	619.8
BIC (smaller is better)	621.4

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	0.13	0.7162

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
treatmnt	1	16	8.08	0.0118
time	4	64	0.98	0.4227
treatmnt*time	4	64	5.85	0.0004

The GLM Procedure

Class Level Information

Class	Levels	Values
treatmnt	2	CONTROL TREATED
time	5	2 4 8 12 16

Number of Observations Read 90
Number of Observations Used 90

The GLM Procedure

Dependent Variable: score

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	9	3254.51659	361.61295	3.68	0.0007
Error	80	7861.03896	98.26299		
Corrected Total	89	11115.55556			

R-Square	Coeff Var	Root MSE	score Mean
0.292789	14.41275	9.912769	68.77778

Source	DF	Type I SS	Mean Square	F Value	Pr > F
treatmnt	1	887.503608	887.503608	9.03	0.0035
time	4	135.000000	33.750000	0.34	0.8478
treatmnt*time	4	2232.012987	558.003247	5.68	0.0004

Source	DF	Type III SS	Mean Square	F Value	Pr > F
treatmnt	1	887.503608	887.503608	9.03	0.0035
time	4	375.346320	93.836580	0.95	0.4369
treatmnt*time	4	2232.012987	558.003247	5.68	0.0004

```
 ID, ConfAmt, Order, ResEater, amtEatC, amtEatA, fullC, fullA, deptimeC
 8,0.0,0.0,1.0,16.00,11.00,2.00,4.00,2.30,4.30
 102,0.0,0.0,1.0,12.00,12.00,1.00,1.00,17.25,5.25
 106,0.0,0.0,2.0,12.00,13.00,1.00,1.00,2.90,3.40
 12,0.0,0.0,1.0,11.00,11.00,1.00,1.00,4.75,7.25
 14,0.0,0.0,1.0,10.00,7.00,1.00,1.00,18.12,5.12
 88,0.0,0.0,2.0,8.50,10.00,3.00,5.00,2.93,2.93
 108,0.0,0.0,1.0,8.00,11.00,1.00,1.00,16.00,16.00
 19,0.0,0.0,2.0,6.00,14.00,1.00,2.00,7.88,7.38


/* foodpsych.sas */
options linesize=79 pagesize=500 noovp formdlim='_';
title 'Social facilitation in eating: ANCOVA with time-varying covariates';
title2 "Patricia Pliner's Eating Norm Study";  /* Ignoring Order for Simplicity */

proc format;
     value ofmt 0 = 'AloneConfederate' 1 = 'ConfederateAlone';
     value amtfmt 0 = 'A little'  1 = 'A lot';
     value combofmt 1 = 'Absent  Little' 2 = 'Absent  Lot'
                    3 = 'Present Little' 4 = 'Present Lot';

data hungry;
     infile eatingnorm.data' firstobs=2 delimiter=',';
     input ID ConfAmt Order ResEater amtEatC amtEatA fullC fullA deptimeC
           deptimeA;
     label
           ConfAmt  = 'Amount eaten by Confederate'
           Order = 'Order'
           amtEatC  = 'Amount eaten with confederate present'
           amtEatA  = 'Amount eaten with confederate absent'
           fullC    = 'Rated fullness before eating, confederate present'
           fullA    = 'Rated fullness before eating, confederate absent'
           deptimeC = 'Deprivation time, confederate present'
           deptimeA = 'Deprivation time, confederate absent';
      format Order ofmt.; format ConfAmt amtfmt.;

data fressen; /* Separate cases for confederate present, absent */
     set hungry;
     confed = 'Absent';
          amtEat = amtEatA; full = fullA; deptime = deptimeA;
          if ConfAmt = 0 then      combo=1;
          else if ConfAmt = 1 then combo=2;
          output;
     confed = 'Present';
          amtEat = amtEatC; full = fullC; deptime = deptimeC;
          if ConfAmt = 0 then      combo=3;
          else if ConfAmt = 1 then combo=4;
          output;
     keep ID ConfAmt Order confed amtEat full deptime combo;
     label
           amtEat   = 'Amount eaten'
           full     = 'Rated fullness before eating'
           deptime  = 'Food Deprivation time';
     format combo combofmt.;
```

```
proc sort;
     by id;

proc print noobs; /* Don't print "observation" (line number) */
     var id confed ConfAmt combo full deptime amtEat ;
     where id le 10; /* Just first 10 cases */

proc freq;
     tables (confed ConfAmt) * combo / norow nocol nopercent missing;

proc mixed;
     class Confed ConfAmt;
     model amtEat = deptime full Confed|ConfAmt;
     repeated / type=un subject=id;
     lsmeans Confed|ConfAmt;


proc mixed;
     class combo;
     model amtEat = deptime full combo;
     repeated / type=un subject=id;
     contrast 'Abs: Lit vs Lot'  combo 1 -1  0  0;
     contrast 'Abs vs PresLit'   combo 1  1 -2  0;
     contrast 'Abs vs PresLot'   combo 1  1  0 -2;
     contrast 'Pres: Lit vs Lot' combo 0  0  1 -1;
     lsmeans combo;
```

Social facilitation in eating: ANCOVA with time-varying covariates 1
Patricia Pliner's Eating Norm Study

ID	confed	ConfAmt	combo	full	deptime	amt Eat
2	Absent	A little	Absent Little	1	2.78	11
2	Presen	A little	Present Little	1	5.78	6
3	Absent	A little	Absent Little	4	3.23	4
3	Presen	A little	Present Little	2	2.73	2
4	Absent	A lot	Absent Lot	1	4.35	11
4	Presen	A lot	Present Lot	1	3.35	16
6	Absent	A lot	Absent Lot	1	7.10	14
6	Presen	A lot	Present Lot	1	1.60	13
7	Absent	A little	Absent Little	2	2.67	9
7	Presen	A little	Present Little	2	5.67	3
8	Absent	A little	Absent Little	4	4.30	11
8	Presen	A little	Present Little	2	2.30	16
9	Absent	A lot	Absent Lot	2	3.95	18
9	Presen	A lot	Present Lot	1	5.95	20
10	Absent	A little	Absent Little	3	3.37	12
10	Presen	A little	Present Little	1	4.37	9

Social facilitation in eating: ANCOVA with time-varying covariates 2
Patricia Pliner's Eating Norm Study

The FREQ Procedure

Table of confed by combo

confed combo

Frequency	Absent Little	Absent Lot	Present Little	Present Lot	Total
Absent	37	37	0	0	74
Presen	0	0	37	37	74
Total	37	37	37	37	148

Table of ConfAmt by combo

ConfAmt(Amount eaten by Confederate) combo

Frequency	Absent Little	Absent Lot	Present Little	Present Lot	Total
A little	37	0	37	0	74
A lot	0	37	0	37	74
Total	37	37	37	37	148

The Mixed Procedure

Model Information

Data Set	WORK.FRESSEN
Dependent Variable	amtEat
Covariance Structure	Unstructured
Subject Effect	ID
Estimation Method	REML
Residual Variance Method	None
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

Class Level Information

Class	Levels	Values
confed	2	Absent Presen
ConfAmt	2	A little A lot

Dimensions

Covariance Parameters	3
Columns in X	11
Columns in Z	0
Subjects	74
Max Obs Per Subject	2

Number of Observations

Number of Observations Read	148
Number of Observations Used	148
Number of Observations Not Used	0

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	890.64916946	
1	2	838.94912902	0.00000034
2	1	838.94903119	0.00000000

Convergence criteria met.

Covariance Parameter Estimates

Cov Parm	Subject	Estimate
UN(1,1)	ID	27.0919
UN(2,1)	ID	18.4257
UN(2,2)	ID	24.2078

Fit Statistics

-2 Res Log Likelihood	838.9
AIC (smaller is better)	844.9
AICC (smaller is better)	845.1
BIC (smaller is better)	851.9

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
2	51.70	<.0001

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
deptime	1	72	0.00	0.9475
full	1	72	3.58	0.0625
confed	1	72	4.46	0.0382
ConfAmt	1	72	10.83	0.0016
confed*ConfAmt	1	72	13.51	0.0005

Least Squares Means

Effect	confed	Amount eaten by Confederate	Estimate	Standard Error	DF	t Value
confed	Absent		10.2728	0.6066	72	16.93
confed	Presen		9.3218	0.5736	72	16.25
ConfAmt		A little	7.9792	0.7766	72	10.27
ConfAmt		A lot	11.6154	0.7766	72	14.96
confed*ConfAmt	Absent	A little	9.2865	0.8714	72	10.66
confed*ConfAmt	Absent	A lot	11.2592	0.8592	72	13.10
confed*ConfAmt	Presen	A little	6.6719	0.8106	72	8.23
confed*ConfAmt	Presen	A lot	11.9716	0.8158	72	14.67

Least Squares Means

Effect	confed	Amount eaten by Confederate	Pr > |t|
confed	Absent		<.0001
confed	Presen		<.0001
ConfAmt		A little	<.0001
ConfAmt		A lot	<.0001
confed*ConfAmt	Absent	A little	<.0001
confed*ConfAmt	Absent	A lot	<.0001
confed*ConfAmt	Presen	A little	<.0001
confed*ConfAmt	Presen	A lot	<.0001

Repeating ...

Least Squares Means

Effect	confed	Amount eaten by Confederate	Estimate
confed*ConfAmt	Absent	A little	9.2865
confed*ConfAmt	Absent	A lot	11.2592
confed*ConfAmt	Presen	A little	6.6719
confed*ConfAmt	Presen	A lot	11.9716



The Mixed Procedure

Model Information

Data Set	WORK.FRESSEN
Dependent Variable	amtEat
Covariance Structure	Unstructured
Subject Effect	ID
Estimation Method	REML
Residual Variance Method	None
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

Class Level Information

Class	Levels	Values
combo	4	Absent Little Absent Lot Present Little Present Lot

Dimensions

Covariance Parameters	3
Columns in X	7
Columns in Z	0
Subjects	74
Max Obs Per Subject	2

Number of Observations

Number of Observations Read	148
Number of Observations Used	148
Number of Observations Not Used	0

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	890.64916946	
1	2	838.94912902	0.00000034
2	1	838.94903119	0.00000000

Convergence criteria met.

Covariance Parameter Estimates

Cov Parm	Subject	Estimate
UN(1,1)	ID	27.0919
UN(2,1)	ID	18.4257
UN(2,2)	ID	24.2078

Fit Statistics

-2 Res Log Likelihood	838.9
AIC (smaller is better)	844.9
AICC (smaller is better)	845.1
BIC (smaller is better)	851.9

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
2	51.70	<.0001

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
deptime	1	73	0.00	0.9475
full	1	73	3.58	0.0625
combo	3	73	10.45	<.0001

Contrasts

Label	Num DF	Den DF	F Value	Pr > F
Abs: Lit vs Lot	1	73	2.56	0.1143
Abs vs PresLit	1	73	24.62	<.0001
Abs vs PresLot	1	73	5.31	0.0241
Pres: Lit vs Lot	1	73	21.13	<.0001

Least Squares Means

Effect	combo	Estimate	Standard Error	DF	t Value	Pr > |t|
combo	Absent Little	9.2865	0.8714	73	10.66	<.0001
combo	Absent Lot	11.2592	0.8592	73	13.10	<.0001
combo	Present Little	6.6719	0.8106	73	8.23	<.0001
combo	Present Lot	11.9716	0.8158	73	14.67	<.0001